Válcové a sférické souřadnice

1. **Transformace skalární veličiny do jiných souřadnic**

V podstatě je o pouhou substituci, nicméně pomoc počítače se může hodit.
Zejména, pokud navazuje výpočet rozvoje pro velká rm jako v následujících příkladech.

Potenciál dipólu ve směru *x*

In[ ]:=
```
Φ = p x / Sqrt[x^2 + y^2 + z^2]^3
$Assumptions = r > 0

TransformedField["Cartesian" → "Spherical", Φ, {x, y, z} → {r, θ, ϕ}] // FullSimplify
```

Out[ ]= $\frac{p\,x}{\left(x^2+y^2+z^2\right)^{3/2}}$

Out[ ]= $r > 0$

Out[ ]= $\frac{p\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]}{r^2}$

Potenciál pro dva bodové náboje na ose z vyjádřený ve válcových souřadnicích

In[ ]:=
```
Φ = q / Sqrt[x^2 + y^2 + (z - a)^2] - q / Sqrt[x^2 + y^2 + (z + a)^2]
$Assumptions = R > 0

TransformedField["Cartesian" → "Cylindrical", Φ, {x, y, z} → {R, ϕ, Z}] // FullSimplify
```

Out[ ]= $\frac{q}{\sqrt{x^2+y^2+(-a+z)^2}}-\frac{q}{\sqrt{x^2+y^2+(a+z)^2}}$

Out[ ]= $R > 0$

Out[ ]= $q\left(\frac{1}{\sqrt{R^2+(a-Z)^2}}-\frac{1}{\sqrt{R^2+(a+Z)^2}}\right)$

2. **Transformace vektorové veličiny do jiných souřadnic**

Zde již nejde o pouhou substituci, je třeba měnit bázová pole

Příklad ze cvičení : trasformace pole ze sférických do kartézských souřadnic.

In[ ]:=
```
f = 2 r^2 LegendreP[2, Cos[θ]]
A = Grad[f, {r, θ, ϕ}, "Spherical"]
```

Out[ ]= $r^2 \left(-1 + 3\,\mathrm{Cos}[\theta]^2\right)$

Out[ ]= $\left\{2\,r \left(-1 + 3\,\mathrm{Cos}[\theta]^2\right),\ -6\,r\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta],\ 0\right\}$

In[ ]:=
```
B = TransformedField["Spherical" → "Cartesian", A, {r, θ, ϕ} → {x, y, z}] // FullSimplify
```

Out[ ]= $\{-2\,x,\ -2\,y,\ 4\,z\}$

Příklad ze cvičení: následuje výpočet divergence v obou souřadnicích

In[ ]:=
```
Div[B, {x, y, z}]
```

Out[ ]= 0

In[ ]:=
```
Div[A, {r, θ, ϕ}, "Spherical"] // Simplify (* to "Spherical" je nebytné *)
```

Out[ ]= 0

Bázový vektor $\boldsymbol{e}_\theta$ vyjádřený v kartézských souřadnicích

In[ ]:=
```
TransformedField["Spherical" → "Cartesian",
  {0, 1, 0}, {r, θ, ϕ} → {x, y, z}] // FullSimplify
```

Out[ ]= $\left\{\frac{x\,z}{\sqrt{x^2+y^2}\ \sqrt{x^2+y^2+z^2}},\ \frac{y\,z}{\sqrt{x^2+y^2}\ \sqrt{x^2+y^2+z^2}},\ -\frac{\sqrt{x^2+y^2}}{\sqrt{x^2+y^2+z^2}}\right\}$

Další příklady

Potenciál sady stejných nábojů ve vrcholech **čtverce** o straně *2a*.

```
In[ ]:= Φ2 = q / Sqrt[(x - a)^2 + (y - a)^2 + z^2] + q / Sqrt[(x + a)^2 + (y - a)^2 + z^2] +
          q / Sqrt[(x - a)^2 + (y + a)^2 + z^2] + q / Sqrt[(x + a)^2 + (y + a)^2 + z^2]
        $Assumptions = r > 0 && a > 0

        TransformedField["Cartesian" → "Spherical", Φ2 , {x, y, z} → {r, θ, ϕ}] // FullSimplify
        Series[%, {r, Infinity, 6}] // FullSimplify
```

Out[ ]=
$$\frac{q}{\sqrt{(-a+x)^2+(-a+y)^2+z^2}}+\frac{q}{\sqrt{(a+x)^2+(-a+y)^2+z^2}}+\frac{q}{\sqrt{(-a+x)^2+(a+y)^2+z^2}}+\frac{q}{\sqrt{(a+x)^2+(a+y)^2+z^2}}$$

Out[ ]= r > 0 && a > 0

Out[ ]=
$$q\left(\frac{\sqrt{2\,a^2+r^2-2\,a\,r\,\text{Sin}[\theta]\,(\text{Cos}[\phi]+\text{Sin}[\phi])}}{\sqrt{4\,a^4+2\,a^2\,r^2+r^4+2\,a^2\,r^2\,(\text{Cos}[2\,\theta]-2\,\text{Sin}[\theta]^2\,\text{Sin}[2\,\phi])}}+\frac{\sqrt{2\,a^2+r^2+2\,a\,r\,\text{Sin}[\theta]\,(\text{Cos}[\phi]+\text{Sin}[\phi])}}{\sqrt{4\,a^4+2\,a^2\,r^2+r^4+2\,a^2\,r^2\,(\text{Cos}[2\,\theta]-2\,\text{Sin}[\theta]^2\,\text{Sin}[2\,\phi])}}+\frac{\sqrt{2\,a^2+r^2+2\,a\,r\,\text{Sin}[\theta]\,(\text{Cos}[\phi]-\text{Sin}[\phi])}}{\sqrt{4\,a^4+2\,a^2\,r^2+r^4+2\,a^2\,r^2\,(\text{Cos}[2\,\theta]+2\,\text{Sin}[\theta]^2\,\text{Sin}[2\,\phi])}}+\frac{\sqrt{2\,a^2+r^2+2\,a\,r\,\text{Sin}[\theta]\,(-\text{Cos}[\phi]+\text{Sin}[\phi])}}{\sqrt{4\,a^4+2\,a^2\,r^2+r^4+2\,a^2\,r^2\,(\text{Cos}[2\,\theta]+2\,\text{Sin}[\theta]^2\,\text{Sin}[2\,\phi])}}\right)$$

Out[ ]=
$$\frac{4\,q}{r}-\frac{a^2\,q\,(1+3\,\text{Cos}[2\,\theta])}{r^3}+\frac{a^4\,q\,(27+60\,\text{Cos}[2\,\theta]+105\,\text{Cos}[4\,\theta]-280\,\text{Cos}[4\,\phi]\,\text{Sin}[\theta]^4)}{32\,r^5}+O\left[\frac{1}{r}\right]^7$$

Potenciál sady stejných nábojů ve vrcholech **krychle** o hraně *2a*.

```
In[ ]:= Φ3 = Sum[q / Sqrt[(x - X)^2 + (y - Y)^2 + (z - Z)^2],
          {X, -a, a, 2 a}, {Y, -a, a, 2 a}, {Z, -a, a, 2 a}]
        $Assumptions = r > 0 && a > 0;

        TransformedField["Cartesian" → "Spherical", Φ3, {x, y, z} → {r, θ, ϕ}] // Simplify;
        Φ3a = Series[%, {r, Infinity, 10}] // FullSimplify
```

Out[ ]=

$$\frac{q}{\sqrt{(-a+x)^2+(-a+y)^2+(-a+z)^2}}+\frac{q}{\sqrt{(a+x)^2+(-a+y)^2+(-a+z)^2}}+\frac{q}{\sqrt{(-a+x)^2+(a+y)^2+(-a+z)^2}}+\frac{q}{\sqrt{(a+x)^2+(a+y)^2+(-a+z)^2}}+\frac{q}{\sqrt{(-a+x)^2+(-a+y)^2+(a+z)^2}}+\frac{q}{\sqrt{(a+x)^2+(-a+y)^2+(a+z)^2}}+\frac{q}{\sqrt{(-a+x)^2+(a+y)^2+(a+z)^2}}+\frac{q}{\sqrt{(a+x)^2+(a+y)^2+(a+z)^2}}$$

Out[ ]=

$$\frac{8\,q}{r}-\frac{7\left(a^4\,q\left(9+20\,\mathrm{Cos}[2\,\theta]+35\,\mathrm{Cos}[4\,\theta]+40\,\mathrm{Cos}[4\,\phi]\,\mathrm{Sin}[\theta]^4\right)\right)}{16\,r^5}+\frac{3\,a^6\,q\left(50+105\,\mathrm{Cos}[2\,\theta]+126\,\mathrm{Cos}[4\,\theta]+231\,\mathrm{Cos}[6\,\theta]-336\times(9+11\,\mathrm{Cos}[2\,\theta])\,\mathrm{Cos}[4\,\phi]\,\mathrm{Sin}[\theta]^4\right)}{32\,r^7}+\frac{1}{16\,384\,r^9}99\,a^8\,q\left(1225+2520\,\mathrm{Cos}[2\,\theta]+2772\,\mathrm{Cos}[4\,\theta]+3432\,\mathrm{Cos}[6\,\theta]+6435\,\mathrm{Cos}[8\,\theta]+448\times(99+156\,\mathrm{Cos}[2\,\theta]+65\,\mathrm{Cos}[4\,\theta])\,\mathrm{Cos}[4\,\phi]\,\mathrm{Sin}[\theta]^4+8320\,\mathrm{Cos}[8\,\phi]\,\mathrm{Sin}[\theta]^8\right)+O\left[\frac{1}{r}\right]^{11}$$

Co to je za divné funkce v tom rozvoji?

In[ ]:=
```
Normal[Φ3a][[2]]
% /. {q → 1, a → 1, r → 1} // Simplify
Maximize[%, {θ, ϕ}][[1]]
SphericalPlot3D[3 × % + %%, {θ, 0, Pi}, {ϕ, 0, 2 Pi},
 PlotRange → Full, PlotPoints → 50, PlotLabel → %%]
```

Out[ ]= $-\frac{7\, a^4\, q\, (9 + 20\, \mathrm{Cos}[2\,\theta] + 35\, \mathrm{Cos}[4\,\theta] + 40\, \mathrm{Cos}[4\,\phi]\, \mathrm{Sin}[\theta]^4)}{16\, r^5}$

Out[ ]= $-\frac{7}{16} \times (9 + 20\, \mathrm{Cos}[2\,\theta] + 35\, \mathrm{Cos}[4\,\theta] + 40\, \mathrm{Cos}[4\,\phi]\, \mathrm{Sin}[\theta]^4)$

Out[ ]= $\frac{56}{3}$

Out[ ]=

$-\frac{7}{16} \times (40 \sin^4(\theta) \cos(4\,\phi) + 20 \cos(2\,\theta) + 35 \cos(4\,\theta) + 9)$



Out[ ]=


-336 sin⁴(θ) (11 cos(2 θ) + 9) cos(4 ϕ) + 105 cos(2 θ) + 126 cos(4 θ) + 231 cos(6 θ
200
0
−200
200
0
−200
−200
0
200

Out[ ]=


$\sin^4(\theta)\,(156\cos(2\,\theta)+65\cos(4\,\theta)+99)\cos(4\,\phi)+2520\cos(2\,\theta)+2772\cos(4\,\theta)$
200
0
−200
00
0
−200
−200
0
200